

DEPARTMENT OF THE INTERIOR

U.S. GEOLOGICAL SURVEY

ENSTA: A Program to Calculate
Orthopyroxene Compositions from X-ray Data

by

1

Carl A. Carlson

and

1

Barry Moring

Open-File Report 85-104

This report is preliminary and has
not been reviewed for conformity with
U.S. Geological Survey editorial standards.

Any use of trade names is for
descriptive purposes only and
does not imply endorsement by the
U. S. Geological Survey.

1

Menlo Park, California

1984

Introduction

This report describes the computer program ENSTA. ENSTA calculates the $\text{Mg} / (\text{Mg} + \text{Fe})$ ratio of orthopyroxenes from X-ray data. The program has the following features:

- 1) Runs on a HP-85 desk top computer
- 2) Reads X-ray diffractograms via a HP 9111A Graphics Tablet
- 3) Uses the X-ray determination curve of Himmelberg and Jackson (1967) to calculate $\text{Mg} / (\text{Mg} + \text{Fe})$ ratio
- 4) For a given sample accepts readings from multiple diffractograms or from multiple readings of the same diffractogram
- 5) For a given sample permits the review of such multiple readings and the rejection of unacceptable values
- 6) Assigns a sample number to each sample and accepts an optional sample identification label
- 7) Prints sample label, sample number, per cent $\text{Mg} / (\text{Mg} + \text{Fe})$ ratio, average location of the (131) orthopyroxene peak with respect to a LiF (111) standard, and standard deviation in percent $\text{Mg} / (\text{Mg} + \text{Fe})$
- 8) Permits the saving of calculations on the HP-85 Tape cartridge

This results in a major savings in time and allows the user to avoid the tedium of making numerous exacting measurements by hand.

Method of Estimating Mg / (Mg + Fe) Compositions

Himmelberg and Jackson (1967) defined an X-ray determination curve for the magnesium and iron content of orthopyroxene in the composition range Mg 48-85 based on the linear variation of 2 theta for the (131) peak. Their regression equation for this relation when using Cu K alpha 1 radiation and the (111) peak of lithium fluoride as an internal standard is:

$$\Delta 2 \theta (131) \text{ orthopyroxene} - (111) \text{ LiF} = 4.112 - 0.009132 \text{ Mg}$$

Their method requires measuring the distance between the (131) and the (111) peaks on each of four scans (two up and two down). ENSTA makes these measurements and calculates the Mg content based on the average of the four replicate measurements and calculates the standard deviation.

Using the Program

Follow these steps to use ENSTA:

- 1) Connect the HP 9111A to the HFIB port on the HP-85
- 2) Turn on the HP-85 and the HP 9111A
- 3) The HP 9111A will sound a signal if it passes its self check and the green Line light will shine
- 4) Insert the Data Cartridge and type: LOAD "ENSTA"
- 5) Press [END LINE] and [RUN]
- 6) The HP 9111A will sound a signal, and its green Digitize light will shine
- 7) The HP-85 screen will clear and display 5 Special Function Key labels:

	LIST		
COMPUTE	EDIT	LABEL	SAVE

- 8) Place the diffractogram on the digitizer in any orientation
- 9) First digitize a point on the center line of one peak. Now digitize two points on the center line of the other peak such that a line so defined would be perpendicular to the long axis of the diffractogram
- 10) The HP-85 will sound a signal as each point is digitized
- 11) When the second peak has been digitized, the distance between (131) and (111) will be calculated on the basis of the distance between the first point digitized and the line defined by the two points on the second peak. The results will be displayed on the HP-85 CRT
- 12) Repeat steps 9 through 11 for each replicate measurement
- 13) At this point the user will want to choose from the options indicated by the Special Function Key labels

LIST: Lists the multiple readings for this sample
EDIT: Cycles through the multiple readings giving the user the option to accept or delete each value

LABEL: Accepts an optional sample label to be included in the printed results of the calculations

SAVE: Saves all results on the Tape Cartridge

COMPUTE: Calculates the $Mg / (Mg + Fe)$ ratio and associated statistics and prints them

Any order of option selection that makes sense will work

- 14) Repeat Steps 8 through 12 as often as needed

Program Source Code

```

10 DIM F(200),L$(32),D(20),X(3),Y(3)
20 M=200 @ M1=20 @L$(32)=" "
30 CLEAR
40 N,N1,N2=0

99 REM * SET UP *
100 ON KEY# 1,"COMPUTE" GOSUB 1000
110 ON KEY# 2," EDIT" GOSUB 2000
120 ON KEY# 3," LABEL" GOSUB 3000
125 ON KEY# 4," SAVE" GOSUB 5000
126 ON KEY# 6,"LIST" GOSUB 8000
130 OUTPUT 706 ;"IN;CN;SF"
140 ON INTR 7 GOSUB 4000
150 OUTPUT 706 ;"IM,4"
160 ENABLE INTR 7;8
170 CLEAR
175 KEY LABEL
180 GOTO 175

999 REM * COMPUTE *
1000 S,S2=0
1010 FOR I=1 TO N1
1020 S=S+D(I) @ S2=S2+D(I)^2
1030 NEXT I
1040 S=S/N1
1050 N=N+1
1060 F(N)=(S/2-4.112)/(-.009132)+.5
1070 PRINT
1072 PRINT L$;" SAMPLE NO. ";N
1074 S3=SQR((S2-(S*N1)^2/N1)/(N1-1)) @ PRINT @ GOSUB 7100
1076 PRINT USING 1077 ; F(N)
1077 IMAGE DDD.DD,"% Mg / (Mg + Fe)"
1080 PRINT USING 1081 ; "PEAK AT: ";S;" INCHES"
1081 IMAGE K,DD.DD,K
1090 PRINT "STANDARD DEVIATION FOR" @ PRINT USING 1091 ; N1;
    "MEASUREMENTS=";S4;"%";S3;"IN."
1091 IMAGE DD,K,D.DD,K,D.DD,K
1095 N1=0
1100 CLEAR @ RETURN

1999 REM * EDIT *
2000 I=0 @ CLEAR @ OFF KEY# 4
2002 I=I+1
2003 IF I>N1 THEN GOTO 2080
2010 DISP "MEAS. NO. ";I
2020 DISP USING "DD.DD" ; D(I)
2030 ON KEY# 2,"OK" GOTO 2002 @ ON KEY# 1,"DELETE" GOTO 2100
    @ ON KEY# 3,"QUIT" GOTO 2080
2031 KEY LABEL
2035 GOTO 2031
2065 CLEAR

```

```

2070 GOTO 2002
2080 CLEAR @ GOSUB 7000 @ RETURN

2099 REM * DELETE *
2100 FOR J=I+1 TO N1
2110 D(J-1)=D(J)
2120 NEXT J
2125 N1=N1-1
2130 GOTO 2003
2199 REM * LABEL *
2999 REM * LABEL *
3000 DISP "ENTER UP TO 32 CHARACTERS"
3010 INPUT L$
3020 CLEAR @ RETURN

3999 REM * INTERRUPT *
4000 STATUS 7,1 ; A
4010 OUTPUT 706 ; "OS"
4030 ENTER 706 ; S1
4040 IF BIT(S1,2) THEN GOSUB 4100
4050 ENABLE INTR 7;8 @ RETURN

4099 REM * GOOD PT *
4100 IF N1>=M1 THEN BEEP @ DISP "TOO MANY RUNS" @ WAIT
      3000 @ N2=0 @ RETURN
4108 N2=N2+1
4109 BEEP 613062/((11*(N2*440)-134/11),.25*440*N2
4110 OUTPUT 706 ; "OD"
4120 ENTER 706 ; X(N2),Y(N2)
4130 X(N2)=X(N2)/1016 @ Y(N2)=Y(N2)/1016
4140 IF N2=3 THEN GOSUB 4200 @ N2=0
4150 RETURN

4199 REM * DIST *
4200 A=(Y(3)-Y(2))/(X(3)-X(2))
4210 B=-1
4220 C=Y(3)-A*X(3)
4230 N1=N1+1
4240 D(N1)=(A*X(1)+B*Y(1)+C)/SQR(A^2+B^2)
4245 D(N1)=ABS(D(N1))
4248 DISP D(N1)
4250 RETURN

4999 REM * SAVE *
5000 DISP "ENTER FILE NAME"
5010 INPUT A$
5020 ON ERROR GOTO 5080
5025 CREATE A$[1,6],INT(8*M/256)+1
5030 ASSIGN# 1 TO A$[1,6]
5040 PRINT# 1 ; P()
5050 ASSIGN# 1 TO *
5060 OFF ERROR
5070 CLEAR @ RETURN
5080 OFF ERROR @ BEEP @ DISP "CAN'T USE ";A$ @ RETURN

```

5090 END

7000 ON KEY# 1,"COMPUTE" GOSUB 1000

7010 ON KEY# 2," EDIT" GOSUB 2000

7020 ON KEY# 3," LABEL" GOSUB 3000

7030 ON KEY# 4," SAVE" GOSUB 5000

7035 ON KEY# 6,"LIST" GOSUB 8000

7040 KEY LABEL

7050 RETURN

7099 REM *CONV SD TO %*

7100 S4=ABS((S/2-4.112)/(-.009132)-((S+S3)/2-4.112)/(-.009132))

7110 RETURN

7999 REM * LIST DATA *

8000 FOR I=1 TO N1

8010 DISP I;D(I)

8020 NEXT I

8030 RETURN

Bibliography

- Himmelberg, G.R. and Jackson E.D., 1967, X-ray determinative curve for some orthopyroxene compositions of composition Mg 48-85 from the Stillwater Complex, Montana, in Geological Survey research, 1967: U.S. Geological Survey Prof. Paper 575-B, p. B 191-B207